AD-A135 086 SHARP PC-1500 POCKET COMPUTER SOLUTIONS TO STATISTICAL 1/1
DATA ANALYSIS OF C..(U) UNIVERSAL ENERGY SYSTEMS INC
DAYTON OH W J PARK OCT 83 AFWAL-TR-83-4096
UNCLASSIFIED F33615-82-C-5001 F/G 9/2 NL

END
DATE FILMED
12-83
DTIC



AFWAL-TR-83-4096

SHARP PC-1500 POCKET COMPUTER SOLUTIONS TO STATISTICAL DATA ANALYSIS OF COMPOSITE MATERIALS OR METALS

WON J. Park
Universal Energy Systems, Inc.
Dayton, OH 45432

Final Report for period June 1983 - October 1983

October 1983

Approved for public release; distribution unlimited.

MATERIALS LABORATORY
AIR FORCE WRIGHT AERONAUTICAL LABORATORIES
AIR FORCE SYSTEMS COMMAND
WRIGHT-PATTERSON AIR FORCE BASE, OHIO 45433

NOTICE

When Government drawings, specifications, or other data are used for any purpose other than in connection with a definitely related Government procurement operation, the United States Government thereby incurs no responsibility nor any obligation whatsoever; and the fact that the Government may have formulated, furnished, or in any way supplied the said drawings, specifications, or other data, is not to be regarded by implication or otherwise as in any manner licensing the holder or any other person or corporation, or conveying any rights or permission to manufacture, use, or sell any patented invention that may in any way be related thereto.

This report has been reviewed by the Office of Public Affairs (ASD/PA) and is releasable to the National Technical Information Service (NTIS). At NTIS, it will be available to the general public, including foreign nations.

This technical report has been reviewed and is approved for publication.

S.W. Tsai

STEPHEN W. TSAI, Chief
Mechanics and Surface Interactions Branch
Nonmetallic Materials Division

FOR THE COMMANDER

F.D. Cherry

FRANKLIN D. CHERRY, Chief
Nonmetallic Materials Division

"If your address has changed, if you wish to be removed from our mailing list, or if the addressee is no longer employed by your organization please notify AFWAL/MLBM , W-PAFB, Ohio 45433 to help us maintain a current mailing list.

Copies of this report should not be returned unless return is required by security considerations, contractual obligations, or notice on a specific document.

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE		READ INSTRUCTIONS BEFORE COMPLETING FORM
1. REPORT NUMBER AFWAL-TR-83-4096	2. GOVT ACCESSION NO.	3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle) Sharp PC-1500 Pocket Computer Solutions to Statistical Data Analysis of Composite Materials or Metals		5. TYPE OF REPORT & PERIOD COVERED Final June 1983 - October 1983 6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s) Won J. Park		8. CONTRACT OR GRANT NUMBER(s) F33615-82-C-5001
9. PERFORMING ORGANIZATION NAME AND ADDRESS Universal Energy Systems, Inc. 4401 Dayton-Xenia Road Dayton, Ohio 45432		10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS P.E. 61102F 2307-P1-16
11. CONTROLLING OFFICE NAME AND ADDRESS Materials Laboratory (AFWAL/MLBM) Air Force Wright Aeronautical Laboratories Wright-Patterson AFB, Ohio 45433		12. REPORT DATE October 1983 13. NUMBER OF PAGES 27
14. MONITORING AGENCY NAME & ADDRESS(if different from Controlling Office)		15. SECURITY CLASS. (of this report) Unclassified 15a. DECLASSIFICATION DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution unlimited

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

These Programs use the language of BASIC and with additional 8KB RAM memories. The computer software contained in this report is theoretical and/or references and in no way reflect Air Force-owned or developed computer software.

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

BASIC Programming
Weibull distribution
A&B - Allowables
Estimations
S-N Curve (Linear regression)

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

This volume contains the description and instructions of the use of Sharp PC-1500 Pocket Computer for statistical data analysis, in Weibull failure distribution setting, of the strength or fatigue life of composite materials or metals. Instant calculations in maximum likelihood estimates, A and B allowables, S-N curve (linear regression) and prediction can be made for practical use.

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

FOREWORD

This report was prepared in the Mechanics and Surface Interactions Branch (AFWAL/MLBM), Nonmetallic Materials Division, Materials Laboratory, Air Force Wright Aeronautical Laboratories, Wright-Patterson AFB, Ohio. The work was performed under Contract F33615-82-C-5001; SB5448-82-C-0076.

The time period covered by this report was from June to October 1983. Dr. Won J. Park was a senior scientist from Universal Energy Systems, Inc. and Professor of Mathematics and Statistics at Wright State University.

Those who want to receive the programmed magnetic cassette tape should contact Stephen W. Tsai, AFWAL/MLBM, Wright-Patterson AFB, Dayton, OHio 45433, Tel: 513-255-3068.



TABLE OF CONTENTS

SECTION I

USER GENERAL INSTRUCTIONS

(1) The program language for PC-1500 Pocket Computer is BASIC and the elementary part of BASIC program instructions are given in the SHARP Instruction manual, which comes with PC-1500 Pocket Computer.

(2) The 8 K RAM should be attached to the computer to increase its capacity. Printer is also required to attach the computer.

(3) The program is called "STAT". This program name is used to load or save the program into cassette tape recorder.

(4) The programs are started in mode RUN by instruction RUN (press the keys [R] [U] [N] and [ENTER]). User is guided through the programs by simple questions. The user types the chosen answer and presses the key [ENTER].

(5) The program is recorded on magnetic cassette tape. To load the program from the cassette recorder to the computer, press [C] [L] [O] [A] [D] ["] [S] [T] [A] [T] ["] and [ENTER].

(6) The STAT-program contains mainly two sub programs, characterized by the computations of (i) Single-level data and (ii) Multi-level data.

(7) For single-level data, the maximum number of data points which can be input is 48 and for multi-level data, it is the same at each level. The capacity for the stress levels input is 7.

(8) The input data value has to be positive number. Otherwise the computation is resulted in error, indicating that an illogical calculation has been made.

(9) The Weibull parameters are estimated by the maximum likelihood method.

SECTION II

CONTENTS OF PROGRAMS

(A) Computation of Single-Level Data:

(1) Data input

(2) Mean and standard deviation

(3) Weibull parameter estimations

(4) Confidence interval and A & B allowables

(5) Plot of Weibull failure distribution

(B) Computation of Multi-Level Data:

(1) Data input

(2) Weibull parameter estimation (pooled)

(3) Plot of failure distributions

(4) S-N curve (Linear regression)

SECTION III

DESCRIPTION OF STATISTICAL METHODS

(A) Single-Level Data

(1) It is assumed that the failure data $X_1, X_2, \ldots, X_n$ of sample size n were generated from a Weibull distribution with parameters α and β, that is, its cumulative distribution function is expressed as

$$F(x) = 1 - \exp\left[-\left(\frac{x}{\beta}\right)^{\alpha}\right], \quad x > 0.$$

This data may be obtained from ultimate strength tests or fatigue test of composite materials or metals. Note that α and β are called the shape and scale parameters respectively.

(2) The mean and standard deviation are computed simply by

$$\text{mean} \quad \bar{x} = \Sigma x$$

$$\text{standard deviation} \quad s = \left[\frac{\Sigma x^2 - (\Sigma x)^2/n}{n-1}\right]^{1/2}.$$

(3) Weibull parameter estimates are obtained by the maximum likelihood method; the scale parameter estimate $\hat{\alpha}$ is the solution of the following M.L. equation (See [3]):

$$\frac{\Sigma x^{\hat{\alpha}} \ell n x}{\Sigma x^{\hat{\alpha}}} - \frac{1}{\hat{\alpha}} - \frac{\Sigma \ell n x}{n} = 0$$

The solution is calculated by using the newton's approximation method. The scale parameter β is estimated by

$$\hat{\beta} = \left[\frac{1}{n} \sum x^{\hat{\alpha}}\right]^{1/\hat{\alpha}}$$

(4) Confidence interval for β and A & B allowables are obtained from the following formulas:

$$95\% \text{ C.I. for } \beta = \hat{\beta} \exp\left[-\ell_{.95}/\hat{\alpha}\right]$$

where $\hat{\alpha}$, $\hat{\beta}$ are M.L.E. of α and β

and $\ell_{.95}$ = Critical values given in [3],

$$A \text{ - allowable} = \beta^{*} \left[\ell n \left(1/.99\right)\right]^{1/\hat{\alpha}}$$

$$B \text{ - allowable} = \beta^{*} \left[\ell n \left(1/.90\right)\right]^{1/\hat{\alpha}}$$

where β^{*} = 95% C.I. for β.

(5) Note that if a random variable X has a Weibull distribution with parameter α and β, then its log transformation T=lnX has the extreme value distribution with parameters $a = 1/\alpha$ and $b = \ln \beta$; i.e. the cumulative distribution of T is

$$F(t) = 1 - \exp\left[-\exp\left(\frac{t-b}{a}\right)\right].$$

The graphing of cumulative failure distribution is performed under (i) without the log transformation or (ii) with the log transformation. The sample data points are plotted using the median rank $p_j=(j-.3)/(n+.4)$.

(B) Mult-Level Data

(1) Under fatigue stress level S_i, i=1,2,...,m, the fatigue life cycles X_{i1}, X_{i2}, , X_{in_i} are observed from the life test, which is theoratically generated from the Weibull distribution with parameters α and β_i. It is also assumed that the shape parameter α is the same for all stress levels.

(2) Weibull parameters are again estimated by the M.L. method. The common shape parameter α is estimated by solving the M.L. equation (See [2]):

$$\sum_i \left[\frac{\sum_j X_{ij}^{\hat{\alpha}} \ln X_{ij}}{\sum_j X_{ij}^{\hat{\alpha}}} \right] - \frac{m}{\hat{\alpha}} - \sum_i \left[\frac{\sum_j \ln X_{ij}}{n_i} \right] = 0$$

The scale parameter β_i (under stress levels S_i) is obtained from

$$\hat{\beta}_i = \left[\frac{1}{n_i} \sum_j X_{ij}^{\hat{\alpha}} \right]^{1/\hat{\alpha}}, \quad i=1,2,...,m$$

(3) The plotting of failure distribution curve is performed exactly the same way as illustrated in (A)-(5). Here only the log transform of the data is used as data points in the plotting.

(4) S-N curve fitting (linear regression) is carried out under the assumption of Weibull regression model (See [1]). The linear equation to fit is

(a) $Y_i=\ln\hat{\beta}_i = a + b \ln S_i$ (classical S-N curve) or

(b) $Y_i = \ln\hat{\beta}_i = a + b. S_i$.

The case (b) may be applied to the rupture testing data.

Because of various sample size n_i of different stress level, we have to use the weighted least square method (See [4]).

The regression parameters a and b are to estimated minimizing

$$Q = \sum_{i=1}^{m} \frac{(Y_i - \hat{Y}_i)^2}{W_i} \quad \text{where } \hat{Y}_i = \hat{a} + \hat{b} \ln S_i \text{ and } W_i = 1/n_i .$$

The following formulas are used in our computations:

$$\text{RSS} = \text{residual sum of square} = \sum_{i=i}^{m} \frac{(Y_i - Y_i)^2}{W_i}$$

$$\text{RMS} = \text{residual mean square} = \text{RSS}/(m-2)$$

$$\text{SSy} = \text{total variation of y} = \sum_{i=i}^{m} \frac{(Y_i - \bar{Y})^2}{W_i}$$

where $\bar{Y}$ = mean of data Y_i

$$\text{correlation coefficient} = \left[1 - \text{RSS}/\text{SS}_Y\right]^{1/2}$$

$$\text{SS}_{reg} = \text{Sum of square for regression} = \text{SS}_Y - \text{RSS}$$

$$\text{F-score for regression} = (m-2)\ \text{SS}_{reg}/\text{RSS}.$$

which has F-distribution with (1, m-2) degree of freedoms.

For a new stress level S, the prediction of β is computed

$$\beta \text{ (predicted)} = \exp(\hat{a} + \hat{b} \ln S) \quad \text{(Case (a))}.$$

or

$$\beta \text{ (predicted)} = \exp(\hat{a} + \hat{b} S) \quad \text{(Case (b))}.$$

SECTION IV

PROGRAM DESCRIPTION

(A) Flow Diagram:



(B) Key Operation Procedure:

Following data are used to illustrate the key operations.

(i) Single-Level Data: T300/934 $(0/45/90/-45)_s$

Ultimate Tensile Strength in MPa

402	430	443	457	466	473	478
484	489	496	503	509	518	

(ii) Multi-Level Data: T300/934 $(0/45/90/-45)_s$

Tension-Tension Fatigue at 3 Levels

1. At level 1 of maximum stress 290 MPa

1 872 851	2 359 300
1 437 800	1 698 100
215 984	864 420

2. At level 2 of maximum stress 345 MPa

87 373	60 912	51 260
11 491	81 571	69 711
59 320	51 848	54 187

3. At level 3 of maximum stress 400 MPa

1 664	7 954	690
1 643	6 430	100
5 705	8 430	1 121
5 329	7 232	5 347

DISPLAY	INPUT	PRINT OUT & REMARKS
	RUN [ENTER]	
MULTI-LEVEL DATA ? (Y/N)	N [ENTER	
HOW MANY DATA N = ?	13 [ENTER]	SINGE-LEVEL DATA
DATA - 1	402 [ENTER]	INPUT DATA N= 13
?		402.00
DATA - 2	430 [ENTER]	430.00
?		443.00
.	.	452.00
.	.	466.00
.	.	473.00
DATA - 13	518 [ENTER]	478.00
?		484.00
		489.00
		496.00
		503.00
		509.00
		518.00
MEAN & ST. DEV. ? (Y/N)	Y [ENTER]	MEAN= 472.92 ST.DEV.= 33....
WEIBULL PARAMETER ESTIMATE		ALPHA= 18.782 BETA= 4.8202E 00
A,B-ALLOWABLE ? (Y/N)	Y [ENTER]	95% C.I.FOR BETA= 423. A-ALLOWABLE = 370.34 B-ALLOWABLE = 431.[illegible]
WEIBULL PLOT ? (Y/N)	Y [ENTER]	
LOG TRANS. OF DATA ? (Y/N)	N [ENTER]	

DISPLAY	INPUT	PRINT OUT & REMARKS
MULTI-LEVEL DATA ? (Y/N)	Y [ENTER]	
NUMBER OF LEVELS =	3 [ENTER]	MULTI-LEVEL DATA
AT LEVEL - 1		NO.OF LEVELS = 3
NO. OF DATA =	6 [ENTER]	LEVEL WITH LARGEST FATIGUE CYCLE IS LEVEL-1
DATA IN CYCLES		
DATA - 1		NO.OF DATA = 6
?	1872851 [ENTER]	1872851 1437800 215984 2359300 1698100 864420
DATA - 2		STR. LEVEL= 290.0
?	1437800 [ENTER]	
.	.	NO.OF DATA = 9
.	.	82323 11491 59320 60912 81521 51848 51260 69211 54187
.	.	
DATA - 6	864420 [ENTER]	
?		STR. LEVEL= 345.0
APPLIED STRESS LEVEL =	290 [ENTER]	
AT LEVEL - 2		NO.OF DATA = 12
NO. OF DATA =	9 [ENTER]	1664 1643 5205 5329 2954 6430 8430 2232 690 100 1121 5342
DATA - 1	87373 [ENTER]	
?		STR. LEVEL= 400.0
DATA - 2	11491 [ENTER]	
?		
.	.	
.	.	
.	.	
DATA - 9	54187 [ENTER]	
?		

DISPLAY	INPUT	PRINT OUT & REMARKS
APPLIED STRESS LEVEL	345 [ENTER]	
AT LEVEL - 3		
NO. OF DATA =	12 [ENTER]	
DATA - 1	1664 [ENTER]	
?		
DATA - ?	1643 [ENTER]	
?		
.	.	
.	.	
.	.	
DATA - 12	5437 [ENTER]	
?		
APPLIED STRESS LEVEL	400 [ENTER]	

DISPLAY	INPUT	PRINT OUT & REMARKS
WEIBULL PARAMER ESTIMATE		POOLED ALPHA= 1.815 BETA-1= 1.5423E 06 BETA-2= 6.161?E 04 BETA-3 5.0738E 03
WEIBULL PLOT ? (Y/N)	Y [ENTER]	 LOG(CYCLE)
S-N (BETA) CURVE ? (Y/N)	Y [ENTER]	
LOG TRANS OF S ? (Y/N)	Y [ENTER]	REGRESSIN LINE] LOG(N= a + b * LOG((N=BETA)
SELECT STRESS LEVEL =	250 [ENTER]	RES. MEAN SQUARE = 1.8520E-01
MORE STRESS LEVEL ? (Y/N)	N [ENTER]	COR. COEFF. = 0.334 F-SCORE FOR REGR. 1360.1 DEG.OF FREEDOM (1, 1) b (SLOPE) = -12.21 a (INTERCEPT) 1.146?E 0? STR. LEVEL= 250.0 BETA PREDICTED= 2.0180E 02

(C) Memory Contents:

MEMORY	DESCRIPTION	MEMORY	DESCRIPTION
(Memory in General)		(Single level data)	
B	Index variable	N	Sample size
C	No. of levels	T	Mean
I	Index variable	S	Standard deviation
J	Index variable	H	95% C.I. for Beta
K	Control variable	P	A - allowable
Z	Control variable	Q	B - allowable
D$	Control variable	B(I)	Table values from [3]
G$	Control variable	(Weibull	parameter estimation)
L$	Control variable	R	*$\Sigma X^{\alpha} \ln X$
C(I)	n_i = sample size of ith level	S	ΣX^{α}
D(I)	S_i = stress level	T	$\Sigma \ln X$
X(J)	x - coordinate	U	$\Sigma X^{\alpha}(\ln X)^2$
Y(J)	y - coordinate	V	$\Sigma \left[{}^{\Sigma \ln X}/_{n_i} \right]$

† expressed in term of memories

* hear X = failure data

MEMORY	DESCRIPTION	MEMORY	DESCRIPTION
G†	W + C/A/A	L	$\Sigma n_i \beta_i S_i$
A	Shape parameter α	M	$\Sigma n_i \beta_i^2$
B(I)	Scale parameter β_i	G	K*T - I*I
Weibull plot		D	Intercept a
N	Sample size	F	Slope b
Z(j)	Data	E	Residual Sum of Square
W	Width of graph	V	Residual Mean Square
E	Plotting Interval in x-coordinate	R	Correlation Coefficient
S	Scale Inticator	S	F-score for regression
T	Scale Inticator	L	New Stress Level
(S-N Curve)		Y	Beta Predicted
T	Σn_i	O	$\Sigma\left[\Sigma X^{\alpha} \ln X / \Sigma X^{\alpha}\right]$
P	$\Sigma \beta_i, \bar{\beta}$	W†	$\Sigma\left[U/S - (R/S)^2\right]$
Q	$\Sigma n_i (\beta_i - \bar{\beta}_i)^2$	F†	O - C/A - V
I	$\Sigma n_i S_i$		
J	$\Sigma n_i \beta_i$		
K	$\Sigma n_i S_i^2$		

† expressed in term of memories

(D) Sample Programs:

1. T300/934 $[0/\pm45/90]_s$, Ultimate Tensile Strength in MPa and Tension-Tension Fatigue Test Data.

SINGE-LEVEL DATA

INPUT DATA N= 13

402.00
430.00
443.00
457.00
466.00
473.00
478.00
484.00
489.00
496.00
503.00
509.00
518.00

MEAN= 472.92
ST.DEV.= 33.22

ALPHA= 18.782
BETA= 4.8702E 02

95% C.I.FOR BETA=
473.11
A-ALLOWABLE =
370.34
B-ALLOWABLE =
419.69



MULTI-LEVEL DATA

NO.OF LEVELS = 3

LEVEL WITH LARGEST FATIGUE CYCLE IS LEVEL-1

NO.OF DATA = 6
1872851
1437800
215984
2359300
1698100
864420
STR. LEVEL= 290.0

NO.OF DATA = 9
87373
11491
59320
60912
81571
51848
51260
69711
54187
STR. LEVEL= 345.0

NO.OF DATA = 12
1664
1643
5705
5329
7954
6430
8430
7232
690
100
1121
5347
STR. LEVEL= 400.0

POOLED ALPHA=1.815
BETA-1= 1.5473E 06
BETA-2= 6.1615E 04
BETA-3= 5.0738E 03



REGRESSIN LINE IS
logN= a + b * logS
(N=BETA)

RES. MEAN SQUARE =
1.0570E-01

COR.COEFF.= 0.999

F-SCORE FOR REGR.=
1360.13
DEG.OF FREEDOM=
(1,1)

b (SLOPE) =
-17.71
a (INTERCEPT) =
1.1460E 02

STR. LEVEL= 250.0
BETA PREDICTED=
2.0180E 07

2. T300/934 $[0/\pm45/90]_s$, Static Tension Strength in MPa and Tension- Tension Fatigue Test Data

SINGE-LEVEL DATA

INPUT DATA N= 20

345.00
376.00
501.00
398.00
399.00
494.00
407.00
412.00
494.00
492.00
412.00
418.00
482.00
478.00
425.00
444.00
465.00
462.00
444.00
456.00

MEAN= 440.20
ST.DEV.= 44.00

ALPHA= 12.313
BETA= 4.5926E 02

95% C.I FOR BETA=
443.82
A-ALLOWABLE =
305.47
B-ALLOWABLE =
369.69



MULTI-LEVEL DATA

NO.OF LEVELS = 6

LEVEL WITH LARGEST FATIGUE CYCLE IS LEVEL-1

NO.OF DATA = 3
2876800
1698100
864420
STR. LEVEL= 42.0

NO.OF DATA = 3
279440
351900
564720
STR. LEVEL= 46.0

NO.OF DATA = 3
26764
58530
51260
STR. LEVEL= 50.0

NO.OF DATA = 3
9888
7798
30520
STR. LEVEL= 54.0

NO.OF DATA = 3
5705
1643
1664
STR. LEVEL= 58.0

NO.OF DATA = 3
8
229
1079
STR. LEVEL= 62.0

POOLED ALPHA=1.591
BETA-1= 1.9226E 06
BETA-2= 4.0931E 05
BETA-3= 4.6744E 04
BETA-4= 1.7874E 04
BETA-5= 3.3384E 03
BETA-6= 5.6960E 02



REGRESSIN LINE IS
logN= a + b * logS
(N=BETA)

RES. MEAN SQUARE =
2.5099E-01

COR.COEFF.= 0.996

F-SCORE FOR REGR.=
534.91
DEG.OF FREEDOM=
(1,4)

b (SLOPE) =
-20.54
a (INTERCEPT) =
9.1403E 01

STR. LEVEL= 38.0
BETA PREDICTED=
1.7252E 07

STR. LEVEL= 34.0
BETA PREDICTED=
1.6956E 08

3. T/300/934, $[0_2/+45_2/90_2]_s$, Tension-Tension Fatigue Test Data

SINGE LEVEL DATA

INPUT DATA N= 20

11491.00
17578.00
40270.00
41200.00
44830.00
51848.00
54187.00
58530.00
59320.00
60912.00
64070.00
69211.00
70049.00
70497.00
71400.00
81571.00
87373.00
116667.00
367644.00
512600.00

MEAN- 92587.40
ST.DEV. 121686.04

ALPHA= 1.071
BETA= 1.0086E 05

95% C.I FOR BETA=
68085.68
A-ALLOWABLE =
928.74
B-ALLOWABLE =
8329.71



MULTI-LEVEL DATA

NO.OF LEVELS = 2

LEVEL WITH LARGEST FATIGUE CYCLE IS LEVEL-1

NO.OF DATA = 20
11491
17578
40270
41200
44830
51848
54187
58530
59320
60912
64070
69211
70049
70497
71400
81571
87373
116667
367644
512600
STR. LEVEL= 50.0

NO.OF DATA = 20
61
100
359
528
690
1121
1588
1643
1664
3558
4930
329
5105
6430
7232
7954
8430
9060
9590
STR. LEVEL= 58.0

POOLED ALPHA=1.035
BETA-1= 9.9195E 04
BETA-2= 4.1177E 03



REGRESSIN LINE IS
logN= a + b * logS
(N=BETA)

b (SLOPE) =
-2.1437E 01
a (INTERCEPT) =
9.5369E 01

STR. LEVEL= 42.0
BETA PREDICTED=
4.1664E 06

STR. LEVEL= 34.0
BETA PREDICTED=
3.8642E 08

(E) Program Listing:

```
  1:"STAT".TEXT :
    COLOR 0
 10:DIM X(50),Y(50
    ),Z(50),A(2,50
    ),B(50),C(2),D
    (2)
 15:INPUT "MULTI-L
    EVEL DATA ? (Y
    /N)";G$
 20:IF G$="Y"GOTO
    600
 25:LPRINT "SINGE-
    LEVEL DATA":LF
    (1)
 28:GOSUB 1190
 30:INPUT "HOW MAN
    Y DATA N=?";C(
    1)
 35:LPRINT "INPUT
    DATA N=";C(1):
    LF (1)
 45:C=1,K=1,Z=1
 50:FOR J=1TO C(1)
 60:PAUSE "DATA "+
    STR$ (J)+" =":
    INPUT A(1,J)
 65:GOSUB 1200
 70:LPRINT A(1,J):
    NEXT J
150:INPUT "MEAN &
    ST.DEV. ? (Y/N
    )";G$:IF G$="N
    "GOTO 700
155:N=C(1):LF (1)
170:T=0,S=0:FOR J=
    1TO N
174:T=T+A(1,J),S=S
    +A(1,J)*A(1,J)
176:NEXT J
178:S=S-T*T/N
180:T=T/N,S=S/(N-1
    )
182:S=√S
183:GOSUB 1200
184:LPRINT "MEAN="
    ;T
186:LPRINT "ST.DEV
    .=";S:LF (1)
200:GOTO 700
321:IF K>1GOTO 351
325:GRAPH :COLOR 1
330:GLCURSOR (25,-
    240):SORGN
335:LINE (-5,0)-(1
    90,0),0,1
337:LINE (190,0)-(
    190,200),0,1
338:LINE (190,200)
    -(0,200),0,1
340:LINE (0,200)-(
    0,-5),0,1
345:W=180/Z(N)
346:LINE (90,0)-(9
    0,-5),0,1
347:LINE (180,0)-(
    180,-5),0,1
348:LINE (0,100)-(
    -5,100),0,1
349:LINE (0,200)-(
    -5,200),0,1
351:FOR J=1TO N
352:LET Y(J)=(J-.3
    )/(N+.4):NEXT
    J
360:FOR J=1TO N
362:X(J)=Z(J)*W
364:Y(J)=Y(J)*200.
    NEXT J
370:FOR J=1TO N
375:LINE (X(J)-2,Y
    (J)-2)-(X(J)+2
    ,Y(J)+2),0,1
376:LINE (X(J)-2,Y
    (J)+2)-(X(J)+2
    ,Y(J)-2),0,1.
    NEXT J
385:IF K>1GOTO 400
390:E=Z(N)/40
400:IF D$="N"GOTO
    480
415:FOR J=0TO 42
420:X(J)=E*J
430:Y(J)=1-EXP (-
    EXP ((X(J)-LN
    B(1))*A))
440:X(J)=X(J)*W,Y(
    J)=Y(J)*200
450:NEXT J:GOTO 49
    0
480:FOR J=0TO 42
485:X(J)=E*J
486:Y(J)=1-EXP (-(
    X(J)/B(1))^A)
487:X(J)=X(J)*W,Y(
    J)=Y(J)*200:
    NEXT J
490:FOR J=1TO 42
492:LINE (X(J-1),Y
    (J-1))-(X(J),Y
    (J)),0,1:NEXT
    J
496:IF Z=2AND K=1
    GOTO 555
497:IF K=CGOTO 500
498:IF K>1GOTO 555
500:GLCURSOR (-25,
    195)
510:LPRINT "1."
520:GLCURSOR (-25,
    95)
530:LPRINT ".5"
540:GLCURSOR (50,-
    20)
545:S=INT (E*400)
550:S=S/10
551:T=S/2
552:LPRINT STR$ (T
    )
553:GLCURSOR (140,
    -20)
554:LPRINT STR$ (S
    )
555:RETURN
558:LF (2)
560:INPUT "A,B-ALL
    OWABLE ? (Y/N)
    ";G$
562:IF G$="N"GOTO
    900
564:GOSUB 1250
565:GOSUB 1200
566:H=B(1)*EXP (-B
    (N)/A)
567:LPRINT "95% C.
    I.FOR BETA=",H
568:P=H*((LN (1/.9
    9))^(1/A))
570:Q=H*((LN (1/.9
    ))^(1/A))
575:LPRINT "A-ALLO
    WABLE = ",P
580:LPRINT "B-ALLO
    WABLE = ",Q
590:GOTO 900
600:Z=2:LF (2)
601:LPRINT "MULTI-
    LEVEL DATA"
602:LF (1)
603:INPUT "NUMBER
    OF LEVELS =";C
```

```
605:GOSUB 1190
606:LPRINT "NO.OF
    LEVELS =";C:LF
    (1)
608.LPRINT "LEVEL
    WITH LARGEST F
    ATIGUE CYCLE I
    S       LEVEL-1
    ";LF (1)
610:FOR I=1TO C
615:PAUSE "AT LEVE
    L-"+STR$ (I)
620:INPUT "NO.OF D
    ATA =";C(I)
624:GOSUB 1190
625:LPRINT "NO.OF
    DATA =";C(I)
627:IF I>1GOTO 630
628:PAUSE "   DATA
     IN CYCLES"
630:FOR J=1TO C(I)
640:PAUSE "DATA-"+
    STR$ (J):INPUT
    A(I,J)
645:GOSUB 1210
650:LPRINT A(I,J):
    NEXT J
663:GOSUB 1197
665:INPUT "APPLIED
     STRESS LEVEL=
    ";D(I)
675:LPRINT "STR. L
    EVEL=";D(I):LF
    (1)
690:NEXT I
700:A=2:BEEP 2
705:PAUSE "WEIBULL
     PARAMETER EST
    IMATES"
710:U=0,W=0,O=0
720:FOR I=1TO C
725:R=0,S=0,T=0,U=
    0
730:FOR J=1TO C(I)
740.R=R+(A(I,J)^A)
    *LN A(I,J)
750:S=S+A(I,J)^A
760:T=T+LN A(I,J)
770.U=U+(A(I,J)^A)
    *LN A(I,J)*LN
    A(I,J)
775:NEXT J
805:W=W+(U/S-(R/S)
    ^2),V=V+T/C(I)
810:O=O+(R/S):NEXT
     I
825.F=O-C/A-U,G=W+
    C/(A*A)
830:IF ABS (F/G)<.
    01GOTO 847
845:A=A-F/G:GOTO 7
    10
847:GOSUB 1185
849:IF Z=1GOTO 852
850:LPRINT "POOLED
     ALPHA=";LF (-
    1)
851:LPRINT A:GOTO
    853
852:LPRINT "ALPHA=
    ";A
853:FOR I=1TO C
855:B(I)=0:NEXT I
860:FOR I=1TO C
863:FOR J=1TO C(I)
865:B(I)=B(I)+(A(I
    ,J)^A):NEXT J
866:B(I)=(B(I)/C(I
    ))^(1/A)
868:GOSUB 1170
870:IF Z=1GOTO 872
871:LPRINT "BETA-"
    +STR$ (I)+"=";
    B(I):GOTO 875
872:LPRINT "BETA="
    ;B(I)
874:IF Z=1GOTO 558
875:NEXT I:LF (1)
900:INPUT "WEIBULL
     PLOT ? (Y/N)"
    ;G$:IF G$="N"
    GOTO 1300
901:IF Z=2LET D$="
    Y":GOTO 904
902:INPUT "LOG TRA
    NS.OF DATA ? (
    Y/N)";D$
904:FOR I=1TO C
905:N=C(I),K=1
906:FOR J=1TO N
907:Z(J)=A(I,J):
    NEXT J
909:IF D$="Y"GOSUB
     1260
910:GOSUB 1005
928:GOSUB 321
940:NEXT I
945:COLOR 0:TEXT
950:LF (1)
955:IF D$="N"GOTO
    965
960.IF Z=2GOTO 962
961:LPRINT "     L
    OG(DATA)":GOTO
    965
962:LPRINT "     LO
    G(CYCLE)"
965:IF Z=1GOTO 15
970:LF (2):GOTO 13
    00
1000:Z(J)=Z(H),Z(
    H)=M:RETURN
1005:FOR J=2TO N
1110:AREAD Z(J)
1120:FOR H=1TO J-
    1
1130:IF Z(J)<Z(H)
    LET M=Z(J).
    GOSUB 1000
1135:NEXT H
1160:NEXT J:
    RETURN
1170:USING "#.###
    #^":RETURN
1185:USING "###.#
    ##":RETURN
1190:USING "###".
    RETURN
1197:USING "#####
    #":RETURN
1200:USING "######
    ##.##":
    RETURN
1210:USING "#####
    ###":RETURN
1250:B(5)=1.107,B
    (6)=.939,B(7
    )=.829,B(8)=
    .751,B(9)=.6
    91,B(10)=.64
    4,B(11)=.605
1251:B(12)=.572,B
    (13)=.544,B(
    14)=.520,B(1
    5)=.499,B(16
    )=.48,B(17)=
    .463,B(18)=.
    447
1252:B(19)=.433,B
    (20)=.421,B(
    21)=.409,B(2
    2)=.398,B(23
    )=.388,B(24)
    =.379,B(25)=
    .37
1253:B(26)=.362,B
    (27)=.354,B(
    28)=.347,B(2
    9)=.34,B(30)
    =.334,B(31)=
    .328,B(32)=.
    323
```

```
1254:B(33)=.317,B
     (34)=.312,B(
     35)=.307,B(3
     6)=.302,B(37
     )=.297,B(38)
     =.293
1255:B(39)=.289,B
     (40)=.285,B(
     41)=.281,B(4
     2)=.278,B(43
     )=.274,B(44)
     =.271
1256:B(45)=.267,B
     (46)=.264,B(
     47)=.261,B(4
     8)=.258,B(49
     )=.255,B(50)
     =.253:RETURN
1260:FOR J=1TO N
1262:Z(J)=LN Z(J)
     :NEXT J
1265:RETURN
1300:IF Z=1GOTO 1
     5
1302:INPUT "S-N(B
     ETA) CURVE ?
      (Y/N)";L$
1303:IF L$="N"
     GOTO 15
1305:INPUT "LOG T
     RANS.OF S ?
     (Y/N)";S$
1306:IF S$="N"
     GOTO 1308
1307:LPRINT "REGR
     ESSIN LINE I
     S logN= a +
     b * logS
      (N=BETA)":
     GOTO 1310
1308:LPRINT "REGR
     ESSIN LINE I
     S logN = a +
      b * S
       (N=BETA)"
1310:T=0,P=0:LF (
     1)
1312:FOR B=1TO C
1315:IF S$="N"
     GOTO 1320
1317:X(B)=LN D(B)
     :GOTO 1324
1320:X(B)=D(B)
1324:Z(B)=LN (B(B
     ))
1326:T=T+C(B):P=P
     +Z(B):NEXT B
1328:I=0:J=0:K=0:
     L=0:M=0:Q=0
1330:P=P/C
1332:FOR B=1TO C
1335:Q=Q+(Z(B)-P)
     *(Z(B)-P)*C(
     B)
1340:I=I+C(B)*X(B
     ):J=J+Z(B)*C
     (B)
1350:K=K+X(B)*X(B
     )*C(B):L=L+X
     (B)*Z(B)*C(B
     )
1360:M=M+Z(B)*Z(B
     )*C(B):NEXT
     B:LF (1)
1410:G=K*T-I*I
1420:D=(J*K-I*L)/
     G
1430:F=(T*L-I*J)/
     G
1440:E=0
1450:FOR B=1TO C
1460:Y(B)=Z(B)-(D
     +F*X(B))
1470:Y(B)=Y(B)*Y(
     B)*C(B)
1480:E=E+Y(B):
     NEXT B
1495:IF C=2GOTO 1
     540
1500:U=E/(C-2)
1505:GOSUB 1170
1510:LPRINT "RES.
      MEAN SQUARE
      =     ";U:LF
     (1)
1520:GOSUB 1185
1522:R=J(1-E/Q)
1523:LPRINT "COR.
     COEFF.=";R:
     LF (1)
1525:GOSUB 1200
1526:S=(Q-E)/U
1527:LPRINT "F-SC
     ORE FOR REGR
     .=     ";S
1530:LPRINT "DEG.
     OF FREEDOM=
                (1,
     "+STR$ (C-2)
     +")":LF (1)
1540:LPRINT " b (
     SLOPE) =
        ";F
1543:GOSUB 1170
1545:LPRINT " a (
     INTERCEPT) =
           ";D:
     LF (1)
1560:GOSUB 1197
1570:INPUT "SELET
      STRESS LEVE
     L=";O
1575:LPRINT "STR.
      LEVEL=";O
1577:IF S$="N"
     GOTO 1580
1578:O=LN O
1580:Z=D+F*O
1590:Y=EXP (Z)
1595:GOSUB 1170
1600:LPRINT "BETA
      PREDICTED=
         ";Y
1610:INPUT "MORE
     STRESS LEVEL
      ? (Y/N)";R$
     :LF (1)
1620:IF R$="Y"
     GOTO 1560
1650:GOTO 15
1700:END
```

SECTION V

CONCLUSION

The description and instruction of the use of Sharp PC-1500 Pocket Computer for the key calculation of statistical data analysis, in Weibull distribution, are presented in this paper. Instant calculation of A and B allowables, S-N curve fitting and prediction can be made for practical use.

REFERENCE

[1] Lawless, J.F., "Statistical Models and Methods for Life Data", John Wiley & Sons, New York, 1982.

[2] Park, W.J., "Percentiles of Pooled Estimates of Weibull Parameter", IEEE Transactions on Reliability, Vol. R-32, No. 1 April 1983, pp. 91-94.

[3] Thoman, D.R., Bain, L.J., And Antle, C.E., "Inferences on the Parameters of the Weibull Distribution", Technometries, Vol. 11, No. 3, August 1969, pp. 445-460.

[4] Weisberg, S., "Applied Linear Regression", John Wiley & Sons, New York, 1980.

* U.S.Government Printing Office: 1983 — 759-062/547


DATE
ILME